Brazilian numbers are so called as they were first formally presented at the 1994 math Olympiad Olimpiada Iberoamericana de Matematica in Fortaleza, Brazil.

Brazilian numbers are defined as:

The set of positive integer numbers where each number N has at least one natural number B where 1 < B < N-1 where the representation of N in base B has all equal digits.



E.G.

 1, 2 & 3 can not be Brazilian; there is no base B that satisfies the condition 1 < B < N-1.
 4 is not Brazilian; 4 in base 2 is 100. The digits are not all the same.
 5 is not Brazilian; 5 in base 2 is 101, in base 3 is 12. There is no representation where the digits are the same.
 6 is not Brazilian; 6 in base 2 is 110, in base 3 is 20, in base 4 is 12. There is no representation where the digits are the same.
 7 is Brazilian; 7 in base 2 is 111. There is at least one representation where the digits are all the same.
 8 is Brazilian; 8 in base 3 is 22. There is at least one representation where the digits are all the same.
 and so on...


All even integers 2P >= 8 are Brazilian because 2P = 2(P-1) + 2, which is 22 in base P-1 when P-1 > 2. That becomes true when P >= 4.

More common: for all all integers R and S, where R > 1 and also S-1 > R, then R*S is Brazilian because R*S = R(S-1) + R, which is RR in base S-1

The only problematic numbers are squares of primes, where R = S. Only 11^2 is brazilian to base 3.

All prime integers, that are brazilian, can only have the digit 1. Otherwise one could factor out the digit, therefore it cannot be a prime number. Mostly in form of 111 to base Integer(sqrt(prime number)). Must be an odd count of 1 to stay odd like primes > 2


Task

Write a routine (function, whatever) to determine if a number is Brazilian and use the routine to show here, on this page;

 the first 20 Brazilian numbers;
 the first 20 odd Brazilian numbers;
 the first 20 prime Brazilian numbers;


See also

 OEIS:A125134 - Brazilian numbers
 OEIS:A257521 - Odd Brazilian numbers
 OEIS:A085104 - Prime Brazilian numbers


on isBrazilian(n)
    repeat with b from 2 to n - 2
        set d to n mod b
        set temp to n div b
        repeat while (temp mod b = d) -- ((temp > 0) and (temp mod b = d))
            set temp to temp div b
        end repeat
        if (temp = 0) then return true
    end repeat
 
    return false
end isBrazilian
 
on isPrime(n)
    if (n < 4) then return (n > 1)
    if ((n mod 2 is 0) or (n mod 3 is 0)) then return false
    repeat with i from 5 to (n ^ 0.5) div 1 by 6
        if ((n mod i is 0) or (n mod (i + 2) is 0)) then return false
    end repeat
 
    return true
end isPrime
 
-- Task code:
on runTask()
    set output to {}
    set astid to AppleScript's text item delimiters
    set AppleScript's text item delimiters to "  "
 
    set collector to {}
    set n to 1
    repeat until ((count collector) is 20)
        if (isBrazilian(n)) then set end of collector to n
        set n to n + 1
    end repeat
    set end of output to "First 20 Brazilian numbers:  " & linefeed & collector
 
    set collector to {}
    set n to 1
    repeat until ((count collector) is 20)
        if (isBrazilian(n)) then set end of collector to n
        set n to n + 2
    end repeat
    set end of output to "First 20 odd Brazilian numbers:  " & linefeed & collector
 
    set collector to {}
    if (isBrazilian(2)) then set end of collector to 2
    set n to 3
    repeat until ((count collector) is 20)
        if (isPrime(n)) and (isBrazilian(n)) then set end of collector to n
        set n to n + 2
    end repeat
    set end of output to "First 20 prime Brazilian numbers:  " & linefeed & collector
 
    set AppleScript's text item delimiters to linefeed
    set output to output as text
    set AppleScript's text item delimiters to astid
    return output
end runTask
 
return runTask()